

## Statistics パッケージによるカーブフィッティング

Statistics パッケージに用意されている NonlinearFit コマンドを使用することで非線形式へのフィッティングが可能になります。この事例では NonlinearFit を使用した非線形フィッティング方法を紹介します。

```
> restart;
```

### フィッティングデータの作成

今回フィッティングの為にデータは非球面レンズの設計式を使用してデータを作成します。関数に含まれるパラメータはそれぞれ以下の意味を持ちます。

R: 曲率半径

K: コーニック(円錐)定数

A~E: 補正項

```
> model := R*x^2/(1+sqrt(1-(1+K)*R^2*x^2)) + A*x^4 + B*x^6 + C*x^8 + D*x^10;
```

$$\text{model} := \frac{R x^2}{1 + \sqrt{1 - R^2 x^2 - R^2 x^2 K}} + A x^4 + B x^6 + C x^8 + D x^{10} \quad (1)$$

式に含まれるパラメータの値は次のようにします。

```
> params := {
R = 1/2.87977, K = -0.64159,
A = 3.148e-4, B = -2.546e-5,
C = -2.814e-6, D = -3.307e-7
};
```

```
> base_eq := eval(model,
params);
```

```
base_eq :=
```

$$\frac{0.3472499540 x^2}{1 + \sqrt{1 - 0.04321798479 x^2}} + 0.0003148 x^4 - 0.00002546 x^6 - 0.000002814 x^8 - 3.307 \cdot 10^{-7} x^{10} \quad (2)$$

関数からフィッティングで使用するためのデータ点を算出します。

```
> pdata := [seq([i, subs(x=i,
base_eq)], i=-1.5..1.5, 0.02)
];
```

```
> Xd := pdata[1..-1, 1];
```


```
> Yd := pdata[1..-1, 2];
```

今回のフィッティングではデータは式から直接算出したデータを使用しております。実際のフィッティングで使用されるデータには計測誤差や製造誤差が含まれているケースがありますので、フィッティングの前に使用されるデータに対してデータの特異値の除去などが必要になる場合があります。

データ点を確認します。

```
> p1 := plots[pointplot]
(pdata, symbol=circle,
symbolsize=10, color=blue,
gridlines=true);
```

```
> p1;
```



## Statistics[NonlinearFit] を使用したフィッティング

フィッティングを実行するとエラーが返されます。  
フィッティングでは複素数を扱うことができずエラーは計算の際に複素数が発生したためとなります。

このような場合、事前に算出可能な変数がある場合は値を求める必要があります。

```
> Statistics[NonlinearFit](model, Xd, Yd, x, output=
  parametervalues);
Error, (in Statistics:-NonlinearFit) complex value encountered
```

今回の場合、R (曲率半径) がデータより算出可能です。  
中心の3点を取り出し、円の方程式に代入し半径を求めます。

```
> Circle_Point := pdata[75..77];
Circle_Point := [[-0.02, 0.00006945034133], [0., 0.], [0.02, 0.00006945034133]] (3)
```

```
> Circle_Eqn := (x-a)^2+(y-b)^2=r^2;
Circle_Eqn := (x-a)^2 + (y-b)^2 = r^2 (4)
```

```
> Circle_Eqs := [seq(subs(x=Circle_Point[i,1], y=Circle_Point[i,2],
  Circle_Eqn), i=1..3)];
Circle_Eqs := [(-0.02-a)^2 + (0.00006945034133-b)^2 = r^2, a^2 + b^2 = r^2, (0.02-a)^2
  + (0.00006945034133-b)^2 = r^2] (5)
```

```
> Circle_Param := solve(Circle_Eqs, [a, b, r]);
Circle_Param := [[a=0., b=2.879790190, r=-2.879790190], [a=0., b
  =2.879790190, r=2.879790190]] (6)
```

半径の逆数を求め曲率半径の値を決定します。

```
> R_val := 1/abs(eval(r, Circle_Param[1]));
R_val := 0.3472475194 (7)
```

求まった曲率半径をモデル式に代入します。

```
> model2 := subs(R=R_val, model);
model2 := 
$$\frac{0.3472475194 x^2}{1 + \sqrt{1 - 0.1205808397 x^2 - 0.1205808397 x^2 K}} + A x^4 + B x^6 + C x^8 + D x^{10}$$
 (8)
```

フィッティングを実行しパラメータを求めます。


```
> NFSol := Statistics[NonlinearFit](model2, Xd, Yd, x, output=
  parametervalues);
NFSol := [K=0.0612723509978834180, A=-0.00335767546817757534, B
  =-0.000351227971071092098, C=-0.0000217420840100445700, D
  =-0.00000563368328192336901] (9)
```

パラメータをモデル式に代入します。

```
> FitEq := subs(NFSol, model2);
FitEq := 
$$\frac{0.3472475194 x^2}{1 + \sqrt{1 - 0.1279691112 x^2}} - 0.00335767546817757534 x^4 - 0.000351227971071092098 x^6 - 0.0000217420840100445700 x^8 - 0.00000563368328192336901 x^{10}$$
 (10)
```

求まった式をデータ点と共にプロットします。

```
> p2 := plot(FitEq, x=-1.5..1.5);
> plots[display](p1, p2);
```




今回求めた値は実際に使用したパラメータとは全くちがいますが、AからDは形状の補正項となり、フィッティング結果の形状はグラフからもわかるように精度の高い結果を求められております。

## 結果の検証

各点での誤差と関数同士の差のプロット

各点での誤差を計算したものと、設計式とフィッティングで得られた結果の差を重ね描きします。

```
> dif_data := [seq([Xd[i], evalf(Yd[i]-subs(x=Xd[i], FitEq))], i=1.
.nops(Xd))]:
> p4 := plots[pointplot](dif_data, symbol=circle, symbolsize=10,
color=green, gridlines=true):
> p5 := plot(base_eq-FitEq, x=-1.5..1.5, numpoints=1000):
> plots[display](p4, p5);
```



統計的な検証

誤差量の絶対値のリストを作成します。

```
> dif_val := map(abs, dif_data[1..-1, 2]):
```

誤差の二乗和

```
> f := x-> x^2:
```

```
> dif_val2 := map(f, dif_val):
```

```
> convert(dif_val2, `+`);
```

2.050914188 10<sup>-13</sup>

(11)

誤差の総和

```
> convert(dif_val, `+`);
```

0.00000455427530

(12)

誤差の平均

```
> Statistics[Mean](dif_val);
```

3.016076358 10<sup>-8</sup>

(13)

誤差の標準偏差(不変分散)

```
> Statistics[StandardDeviation](dif_val);
```

2.124947031 10<sup>-8</sup>

(14)

## 使用環境

OS Windows XP SP2

CPU Intel Core 2 Duo T7600 2.33GHz

RAM 2GB

Maple 11.0 (Build ID: 277698) GlobalOptimization Toolbox  
ワークシートの一括実行に要した時間：約 0.3 秒  
(時間はあくまでも目安です。その時の PC の状態により変化します。)